In[164]:=
```
(*Define the vertices of the reference triangle*)
vertices = {{0, 0}, {1, 0}, {0, 1}};
```

In[165]:=
```
(*Function to integrate a polynomial over the reference triangle*)
IntegrateOverTriangle[poly_, x_, y_] := Integrate[poly, {x, 0, 1}, {y, 0, 1 - x}];
```

In[166]:=
```
(*Define a polynomial in x and y*)
poly = x^3 + y^2 * x + x * y;
(*Integrate the polynomial over the triangle*)
result = IntegrateOverTriangle[poly, x, y];
Print["Integral of the polynomial over the triangle: ", result];
```

Integral of the polynomial over the triangle: $\frac{13}{120}$

In[169]:=
```
(*Define the six points in counterclockwise order*)
dofs = {{0, 0}, {1 / 2, 0}, {1, 0}, {1 / 2, 1 / 2}, {0, 1}, {0, 1 / 2} };
ListPlot[dofs, PlotStyle → {Red, PointSize[Medium]},
 AxesLabel → {"x", "y"}, GridLines → Automatic]
```

Out[170]=



In[171]:=
```
(*Function to evaluate a polynomial at a list of points*)
EvaluatePolynomialOnDofs[poly_, x_, y_, dofs_] :=
  Table[poly /. {x → point[[1]], y → point[[2]]}, {point, dofs}];
```

In[172]:=
```
EvaluatePolynomialOnDofs[poly, x, y, dofs]
```

Out[172]= $\left\{0, \frac{1}{8}, 1, \frac{1}{2}, 0, 0\right\}$

In[173]:=
```
(*Define the general form of a degree 2 polynomial in x and y*)
genericPoly[x_, y_] := a x^2 + b x y + c y^2 + d x + e y + f;
```

In[174]:=
```
CreateEquations[targetDOF_] :=
  Table[genericPoly[dofs[[i, 1]], dofs[[i, 2]]] == If[i == targetDOF, 1, 0], {i, 1, 6}];
```

In[175]:=
```
basisPolynomials = Table[variables = {a, b, c, d, e, f};
   equations = CreateEquations[i];
   solution = Solve[equations, variables];
   poly = genericPoly[x, y] /. solution[[1]];
   Simplify[poly], {i, 1, 6}];
```

In[176]:=
```
basisPolynomials
```
Out[176]=
$$\left\{1+2x^2-3y+2y^2+x(-3+4y),\ -4x(-1+x+y),\ x(-1+2x),\ 4xy,\ y(-1+2y),\ -4y(-1+x+y)\right\}$$

In[177]:=
```
Do[Print["Basis polynomial ", i, ": ", basisPolynomials[[i]]], {i, 1, 6}]
```
```
Basis polynomial 1: 1 + 2 x^2 - 3 y + 2 y^2 + x (-3 + 4 y)
Basis polynomial 2: -4 x (-1 + x + y)
Basis polynomial 3: x (-1 + 2 x)
Basis polynomial 4: 4 x y
Basis polynomial 5: y (-1 + 2 y)
Basis polynomial 6: -4 y (-1 + x + y)
```

In[178]:=
```
Do[values = EvaluatePolynomialOnDofs[basisPolynomials[[i]], x, y, dofs];
 Print["Verification for polynomial ", i, ": ", values];, {i, 1, 6}]
```
```
Verification for polynomial 1: {1, 0, 0, 0, 0, 0}
Verification for polynomial 2: {0, 1, 0, 0, 0, 0}
Verification for polynomial 3: {0, 0, 1, 0, 0, 0}
Verification for polynomial 4: {0, 0, 0, 1, 0, 0}
Verification for polynomial 5: {0, 0, 0, 0, 1, 0}
Verification for polynomial 6: {0, 0, 0, 0, 0, 1}
```

In[179]:=
```
Plot3D[basisPolynomials[[3]], {x, 0, 1}, {y, 0, 1}]
```
Out[179]=



In[180]:=
```
Fx[x_, y_] := xa + (xb - xa) * x + (xc - xa) * y
Fy[x_, y_] := ya + (yb - ya) * x + (yc - ya) * y
```

In[182]:=
```
JF = Simplify[{{D[Fx[x, y], x], D[Fx[x, y], y]}, {D[Fy[x, y], x], D[Fy[x, y], y]}}]
```
Out[182]=
$$\{\{-xa+xb,\ -xa+xc\},\ \{-ya+yb,\ -ya+yc\}\}$$

In[183]:=
```
den = (xc * (-ya + yb) + xb (ya - yc) + xa (-yb + yc)) ^ 2
jacobianFactor = FullSimplify[den * Inverse[JF].Transpose[Inverse[JF]]]
```

Out[183]= $(xc\,(-ya+yb)+xb\,(ya-yc)+xa\,(-yb+yc))^2$

Out[184]= $\left\{\left\{(xa-xc)^2+(ya-yc)^2,\ -((xa-xb)\,(xa-xc))-(ya-yb)\,(ya-yc)\right\},\right.$
$\left.\left\{-((xa-xb)\,(xa-xc))-(ya-yb)\,(ya-yc),\ (xa-xb)^2+(ya-yb)^2\right\}\right\}$

In[185]:=
```
integrand[i_, j_] := Dot[jacobianFactor.Grad[basisPolynomials[[i]], {x, y}],
  Grad[basisPolynomials[[j]], {x, y}]]
```

In[186]:=
```
stiffness[i_, j_] := IntegrateOverTriangle[integrand[i, j], x, y]
```

In[187]:=
```
stiffnessMatrix = ParallelTable[stiffness[i, j], {i, 1, 6}, {j, 1, 6}];
MatrixForm[stiffnessMatrix]
```

Out[188]//MatrixForm=

$$\begin{pmatrix}
\frac{1}{2}\left((xb-xc)^2+(yb-yc)^2\right) & -\frac{2}{3}\left(xa\,(xb-xc)-xb\,xc\right. \\
-\frac{2}{3}\left(xa\,(xb-xc)-xb\,xc+xc^2+ya\,yb-ya\,yc-yb\,yc+yc^2\right) & \frac{4}{3}\left(xa^2+xb^2-xb\,xc+xc^2-xa\,(xb+\right. \\
\frac{1}{6}\left((xa-xc)\,(xb-xc)+(ya-yc)\,(yb-yc)\right) & -\frac{2}{3}\left((xa-xc)\,(xb\right. \\
0 & \frac{4}{3}\left(-xb^2+xa\,(xb-xc\right. \\
\frac{1}{6}\left(-((xa-xb)\,(xb-xc))-(ya-yb)\,(yb-yc)\right) & \\
\frac{2}{3}\left((xa-xb)\,(xb-xc)+(ya-yb)\,(yb-yc)\right) & -\frac{4}{3}\left(xa^2+xb\,xc-xa\,(\right.
\end{pmatrix}$$

In[189]:=
```
Export["P2.csv", stiffnessMatrix, "CSV"]
```

Out[189]= P2.csv

In[190]:=
```
shapeIntegralOnRef =
 Table[IntegrateOverTriangle[basisPolynomials[[i]], x, y], {i, 1, 6}]
```

Out[190]= $\left\{0,\ \frac{1}{6},\ 0,\ \frac{1}{6},\ 0,\ \frac{1}{6}\right\}$

In[191]:=