In[192]:=
```
(*Define the vertices of the reference triangle*)
vertices = {{0, 0}, {1, 0}, {0, 1}};
```

In[193]:=
```
(*Function to integrate a polynomial over the reference triangle*)
IntegrateOverTriangle[poly_, x_, y_] := Integrate[poly, {x, 0, 1}, {y, 0, 1 - x}];
```

In[194]:=
```
poly = x^2 + y^2 + x * y;
(*Integrate the polynomial over the triangle*)
result = IntegrateOverTriangle[poly, x, y];
Print["Integral of the polynomial over the triangle: ", result];
```

Integral of the polynomial over the triangle: $\frac{5}{24}$

In[197]:=
```
dofs = {{0, 0}, {1 / 3, 0}, {2 / 3, 0}, {1, 0}, {2 / 3, 1 / 3},
   {1 / 3, 2 / 3}, {0, 1}, {0, 2 / 3}, {0, 1 / 3}, {1 / 3, 1 / 3}};
ListPlot[dofs, PlotStyle → {Red, PointSize[Medium]},
 AxesLabel → {"x", "y"}, GridLines → Automatic]
```

Out[198]=



In[199]:=
```
(*Function to evaluate a polynomial at a list of points*)
EvaluatePolynomialOnDofs[poly_, x_, y_, dofs_] :=
  Table[poly /. {x → point[[1]], y → point[[2]]}, {point, dofs}];
```

In[200]:=
```
poly = x^3 + x * y^2 + 2 x + 1;
EvaluatePolynomialOnDofs[poly, x, y, dofs]
```

Out[201]= $\left\{1, \frac{46}{27}, \frac{71}{27}, 4, \frac{73}{27}, \frac{50}{27}, 1, 1, 1, \frac{47}{27}\right\}$

In[202]:=
```
genericPoly[x_, y_] :=
 a1 x^3 + a2 y^3 + a3 x^2 y + a4 x y^2 + a5 x^2 + a6 y^2 + a7 x y + a8 x + a9 y + a10
```

In[203]:=
```
CreateEquations[targetDOF_] :=
  Table[genericPoly[dofs[[i, 1]], dofs[[i, 2]]] == If[i == targetDOF, 1, 0], {i, 1, 10}];
```

In[204]:=
```
basisPolynomials = Table[variables = {a1, a2, a3, a4, a5, a6, a7, a8, a9, a10};
  equations = CreateEquations[i];
  solution = Solve[equations, variables];
  poly = genericPoly[x, y] /. solution[[1]];
  Simplify[poly], {i, 1, 10}]
```

Out[204]=
$$\left\{\frac{1}{2}\left(2-9x^3-11y+18y^2-9y^3-9x^2(-2+3y)+x\left(-11+36y-27y^2\right)\right),\ \frac{9}{2}x\left(2+3x^2-5y+3y^2+x(-5+6y)\right),\ -\frac{9}{2}x(-1+3x)(-1+x+y),\ \frac{1}{2}x\left(2-9x+9x^2\right),\ \frac{9}{2}x(-1+3x)y,\ \frac{9}{2}xy(-1+3y),\ \frac{1}{2}y\left(2-9y+9y^2\right),\ -\frac{9}{2}y(-1+x+y)(-1+3y),\ \frac{9}{2}y\left(2+3x^2-5y+3y^2+x(-5+6y)\right),\ -27xy(-1+x+y)\right\}$$

In[205]:=
```
Do[Print["Basis polynomial ", i, ": ", basisPolynomials[[i]]], {i, 1, 10}]
```

Basis polynomial 1: $\frac{1}{2}\left(2-9x^3-11y+18y^2-9y^3-9x^2(-2+3y)+x\left(-11+36y-27y^2\right)\right)$

Basis polynomial 2: $\frac{9}{2}x\left(2+3x^2-5y+3y^2+x(-5+6y)\right)$

Basis polynomial 3: $-\frac{9}{2}x(-1+3x)(-1+x+y)$

Basis polynomial 4: $\frac{1}{2}x\left(2-9x+9x^2\right)$

Basis polynomial 5: $\frac{9}{2}x(-1+3x)y$

Basis polynomial 6: $\frac{9}{2}xy(-1+3y)$

Basis polynomial 7: $\frac{1}{2}y\left(2-9y+9y^2\right)$

Basis polynomial 8: $-\frac{9}{2}y(-1+x+y)(-1+3y)$

Basis polynomial 9: $\frac{9}{2}y\left(2+3x^2-5y+3y^2+x(-5+6y)\right)$

Basis polynomial 10: $-27xy(-1+x+y)$

In[206]:=
```
Do[values = EvaluatePolynomialOnDofs[basisPolynomials[[i]], x, y, dofs];
 Print["Verification for polynomial ", i, ": ", values];, {i, 1, 10}]
```

```
Verification for polynomial 1: {1, 0, 0, 0, 0, 0, 0, 0, 0, 0}
Verification for polynomial 2: {0, 1, 0, 0, 0, 0, 0, 0, 0, 0}
Verification for polynomial 3: {0, 0, 1, 0, 0, 0, 0, 0, 0, 0}
Verification for polynomial 4: {0, 0, 0, 1, 0, 0, 0, 0, 0, 0}
Verification for polynomial 5: {0, 0, 0, 0, 1, 0, 0, 0, 0, 0}
Verification for polynomial 6: {0, 0, 0, 0, 0, 1, 0, 0, 0, 0}
Verification for polynomial 7: {0, 0, 0, 0, 0, 0, 1, 0, 0, 0}
Verification for polynomial 8: {0, 0, 0, 0, 0, 0, 0, 1, 0, 0}
Verification for polynomial 9: {0, 0, 0, 0, 0, 0, 0, 0, 1, 0}
Verification for polynomial 10: {0, 0, 0, 0, 0, 0, 0, 0, 0, 1}
```

In[207]:=
```
Plot3D[basisPolynomials[[3]], {x, 0, 1}, {y, 0, 1}]
```

Out[207]=



In[208]:=
```
Fx[x_, y_] := xa + (xb - xa) * x + (xc - xa) * y
Fy[x_, y_] := ya + (yb - ya) * x + (yc - ya) * y
JF = Simplify[{{D[Fx[x, y], x], D[Fx[x, y], y]}, {D[Fy[x, y], x], D[Fy[x, y], y]}}]
```

Out[210]= {{-xa + xb, -xa + xc}, {-ya + yb, -ya + yc}}

In[211]:=
```
den = (xc * (-ya + yb) + xb (ya - yc) + xa (-yb + yc)) ^2
jacobianFactor = FullSimplify[den * Inverse[JF].Transpose[Inverse[JF]]]
```

Out[211]= $(xc\ (-ya + yb) + xb\ (ya - yc) + xa\ (-yb + yc))^2$

Out[212]= $\{\{(xa - xc)^2 + (ya - yc)^2, -((xa - xb)\ (xa - xc)) - (ya - yb)\ (ya - yc)\},$
$\{-((xa - xb)\ (xa - xc)) - (ya - yb)\ (ya - yc), (xa - xb)^2 + (ya - yb)^2\}\}$

In[213]:=
```
integrand[i_, j_] := Dot[jacobianFactor.Grad[basisPolynomials[[i]], {x, y}],
  Grad[basisPolynomials[[j]], {x, y}]]
stiffness[i_, j_] := IntegrateOverTriangle[integrand[i, j], x, y]
```

In[215]:=
```
stiffnessMatrix = ParallelTable[stiffness[i, j], {i, 1, 10}, {j, 1, 10}];
MatrixForm[stiffnessMatrix]
```

Out[216]//MatrixForm=

$$\begin{pmatrix}
\frac{17}{40}\left((xb-xc)^2+(yb-yc)^2\right) \\
\frac{3}{80}\left(xb^2-19\,xa\,(xb-xc)+17\,xb\,xc-18\,xc^2-19\,ya\,yb+yb^2+19\,ya\,yc+17\,yb\,yc-18\,yc^2\right) \\
\frac{3}{80}\left(xb^2+8\,xa\,(xb-xc)-10\,xb\,xc+9\,xc^2+8\,ya\,yb+yb^2-8\,ya\,yc-10\,yb\,yc+9\,yc^2\right) \quad -\frac{2}{8} \\
-\frac{7}{80}\left((xa-xc)\,(xb-xc)+(ya-yc)\,(yb-yc)\right) \\
-\frac{3}{80}\left((xb-xc)^2+(yb-yc)^2\right) \\
-\frac{3}{80}\left(xb^2-2\,xb\,xc+xc^2+(yb-yc)^2\right) \\
\frac{7}{80}\left((xa-xb)\,(xb-xc)+(ya-yb)\,(yb-yc)\right) \\
\frac{3}{80}\left(-8\,xa\,xb+9\,xb^2+8\,xa\,xc-10\,xb\,xc+xc^2-8\,ya\,yb+9\,yb^2+8\,ya\,yc-10\,yb\,yc+yc^2\right) \\
\frac{3}{80}\left(-18\,xb^2+19\,xa\,(xb-xc)+17\,xb\,xc+xc^2+19\,ya\,yb-18\,yb^2-19\,ya\,yc+17\,yb\,yc+yc^2\right) \\
0
\end{pmatrix}$$

In[217]:=
```
Export["P3.csv", stiffnessMatrix, "CSV"]
```

Out[217]= P3.csv

In[218]:=
```
shapeIntegralOnRef =
 Table[IntegrateOverTriangle[basisPolynomials[[i]], x, y], {i, 1, 10}]
```

Out[218]= $\left\{\frac{1}{60}, \frac{3}{80}, \frac{3}{80}, \frac{1}{60}, \frac{3}{80}, \frac{3}{80}, \frac{1}{60}, \frac{3}{80}, \frac{3}{80}, \frac{9}{40}\right\}$